package com.jinher.lib
{
	/**
 	 * 创建人：王曙光
 	 * 创建时间：2011-10-14
 	 * 功能描述：URL操作相关
 	 */
	public final class URL
	{
		
		//利用BASE64加密解密字符串		 
		private var base64EncodeChars:String="ABCDEFGHIJKLMNOPQRSTUVWXYZabcdefghijklmnopqrstuvwxyz0123456789+/";
		private var base64DecodeChars:Array=new Array(-1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, 62, -1, -1, -1, 63, 52, 53, 54, 55, 56, 57, 58, 59, 60, 61, -1, -1, -1, -1, -1, -1, -1, 0, 1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15, 16, 17, 18, 19, 20, 21, 22, 23, 24, 25, -1, -1, -1, -1, -1, -1, 26, 27, 28, 29, 30, 31, 32, 33, 34, 35, 36, 37, 38, 39, 40, 41, 42, 43, 44, 45, 46, 47, 48, 49, 50, 51, -1, -1, -1, -1, -1);

		//字符串由utf8到utf16
		private function utf8to16(str:String):String
		{
			var out:String;
			var i:int;
			var len:int;
			var c:int;
			var char2:int;
			var char3:int;

			out="";
			len=str.length;
			i=0;
			while (i < len)
			{
				c=str.charCodeAt(i++);
				switch (c >> 4)
				{
					case 0:
					case 1:
					case 2:
					case 3:
					case 4:
					case 5:
					case 6:
					case 7:
						// 0xxxxxxx
						out+=str.charAt(i - 1);
						break;
					case 12:
					case 13:
						// 110x xxxx   10xx xxxx
						char2=str.charCodeAt(i++);
						out+=String.fromCharCode(((c & 0x1F) << 6) | (char2 & 0x3F));
						break;
					case 14:
						// 1110 xxxx  10xx xxxx  10xx xxxx
						char2=str.charCodeAt(i++);
						char3=str.charCodeAt(i++);
						out+=String.fromCharCode(((c & 0x0F) << 12) | ((char2 & 0x3F) << 6) | ((char3 & 0x3F) << 0));
						break;
				}
			}

			return out;
		}

		//字符串由utf16到utf8
		private function utf16to8(str:String):String
		{
			var out:String;
			var i:int;
			var len:int;
			var c:int;

			out="";
			len=str.length;
			for (i=0; i < len; i++)
			{
				c=str.charCodeAt(i);
				if ((c >= 0x0001) && (c <= 0x007F))
				{
					out+=str.charAt(i);
				}
				else if (c > 0x07FF)
				{
					out+=String.fromCharCode(0xE0 | ((c >> 12) & 0x0F));
					out+=String.fromCharCode(0x80 | ((c >> 6) & 0x3F));
					out+=String.fromCharCode(0x80 | ((c >> 0) & 0x3F));
				}
				else
				{
					out+=String.fromCharCode(0xC0 | ((c >> 6) & 0x1F));
					out+=String.fromCharCode(0x80 | ((c >> 0) & 0x3F));
				}
			}
			return out;
		}
		
		/**
		 * 加密字符串,如字符串:1234567890
		 * Returns：MTIzNDU2Nzg5MA==
		 */		
		public function urlEncode(str:String):String
		{
			str=utf16to8(str);
			var out:String;
			var i:int;
			var len:int;
			var c1:int;
			var c2:int;
			var c3:int;

			len=str.length;
			i=0;
			out="";
			while (i < len)
			{
				c1=str.charCodeAt(i++) & 0xff;
				if (i == len)
				{
					out+=base64EncodeChars.charAt(c1 >> 2);
					out+=base64EncodeChars.charAt((c1 & 0x3) << 4);
					out+="==";
					break;
				}
				c2=str.charCodeAt(i++);
				if (i == len)
				{
					out+=base64EncodeChars.charAt(c1 >> 2);
					out+=base64EncodeChars.charAt(((c1 & 0x3) << 4) | ((c2 & 0xF0) >> 4));
					out+=base64EncodeChars.charAt((c2 & 0xF) << 2);
					out+="=";
					break;
				}
				c3=str.charCodeAt(i++);
				out+=base64EncodeChars.charAt(c1 >> 2);
				out+=base64EncodeChars.charAt(((c1 & 0x3) << 4) | ((c2 & 0xF0) >> 4));
				out+=base64EncodeChars.charAt(((c2 & 0xF) << 2) | ((c3 & 0xC0) >> 6));
				out+=base64EncodeChars.charAt(c3 & 0x3F);
			}
			out=out.replace("+", "%2B")
			return out + "&";
		}

		/**
		 * 解密字符串,如字符串:MTIzNDU2Nzg5MA==
		 * Returns：1234567890
		 */		
		public function urlDecode(str:String):String
		{
			str=str.substring(0, str.length - 1);
			var re:RegExp=/%2B/g; // 创建正则表达式模式。
			str=str.replace(re, "+");
			var c1:int;
			var c2:int;
			var c3:int;
			var c4:int;
			var i:int;
			var len:int;
			var out:String;

			len=str.length;
			i=0;
			out="";
			while (i < len)
			{
				/* c1 */
				do
				{
					c1=int(base64DecodeChars[str.charCodeAt(i++) & 0xff]);
				} while (i < len && c1 == -1);
				if (c1 == -1)
					break;
				/* c2 */
				do
				{
					c2=int(base64DecodeChars[str.charCodeAt(i++) & 0xff]);
				} while (i < len && c2 == -1);
				if (c2 == -1)
					break;
				out+=String.fromCharCode((c1 << 2) | ((c2 & 0x30) >> 4));
				/* c3 */
				do
				{
					c3=str.charCodeAt(i++) & 0xff;
					if (c3 == 61){					
        　				return  utf8to16(out);
					}
					c3=int(base64DecodeChars[c3]);
				} while (i < len && c3 == -1);
				if (c3 == -1)
					break;

				out+=String.fromCharCode(((c2 & 0XF) << 4) | ((c3 & 0x3C) >> 2));

				/* c4 */
				do
				{
					c4=str.charCodeAt(i++) & 0xff;
					if (c4 == 61)
					{
						out=utf8to16(out);
						return out;
					}
					c4=int(base64DecodeChars[c4]);
				} while (i < len && c4 == -1);
				if (c4 == -1)
					break;
				out+=String.fromCharCode(((c3 & 0x03) << 6) | c4);
			}
			out=utf8to16(out)
			return out;
		}	
	}
}
